
The Speedup Theorem in a Primitive Recursive Framework

Andrea Asperti
Department of Computer Science and Engineering DISI, University of Bologna

andrea.asperti@unibo.it

Abstract
Blum’s speedup theorem is a major theorem in computational com-
plexity, showing the existence of computable functions for which
no optimal program can exist: for any speedup function r there ex-
ists a function fr such that for any program computing fr we can
find an alternative program computing it with the desired speedup
r. The main corollary is that algorithmic problems do not have, in
general, a inherent complexity.

Traditional proofs of the speedup theorem make an essential
use of Kleene’s fix point theorem to close a suitable diagonal
argument. As a consequence, very little is known about its validity
in subrecursive settings, where there is no universal machine, and
no fixpoints. In this article we discuss an alternative, formal proof
of the speedup theorem that allows us to spare the invocation of the
fix point theorem and sheds more light on the actual complexity of
the function fr .

Categories and Subject Descriptors F.3.1 [Specifying and Ver-
ifying and Reasoning about Programs]: Mechanical Verification;
D.2.4 [Software/Program Verification]: Correctness proofs; F.1.3
[Complexity Measures and Classes]: Complexity hierarchies, Ma-
chine independent complexity; F.3.3 [Studies of Program Con-
structs]: Type Structure, Program and recursion schemes; F.4.1
[Mathematical Logic]: Lambda calculus and related systems, re-
cursive function theory

Keywords Speedup, Primitive Recursion, Machine independent
complexity, Matita

1. Introduction
Each computational problem can be solved by an infinite number
of different programs. Given some complexity measure, counting
the amount of computational resources (such as time or space)
required by the different computations, one is obviously interested
to find, if possible, an optimal program, that is a program with
minimal complexity. Blum’s speedup theorem [11] proves that, no
matter how complexity is measured, there are problems admitting
no optimal solution. As a consequence, a computable function does
not have in general a inherent computational complexity (unless
expressed as an aggregate of the complexities of all its programs).

The speedup theorem is traditionally expressed in Blum’s ab-
stract framework [10]. We write f(x) ↓ to express that the partial
function f is defined for input x, and f(x)↑ otherwise.

DEFINITION 1. (Blum [10]) A pair 〈ϕ,Φ〉 is a computational com-
plexity measure if ϕ is a principal effective enumeration of all par-
tial recursive functions and Φ satisfies the following axioms:

(a) ϕi(~n) ↓↔ Φi(~n) ↓
(b) the predicate Φi(~n) = m is decidable

As traditional in this setting, we adopt the convention that Φi(~n) =
∞ if Φi(~n) ↑; in particular, the relation Φi(~n) > m also holds
when Φi(~n) is undefined. We use the acronym a.e. as an abbre-
viation of almost everywhere, that is for all but a finite number of
values.

THEOREM 2. (Speedup theorem [Blum [11]]) For any speedup
function r there exists a computable function f such that for any
ϕi = f there exists ϕj = f such that

r(Φj(x)) ≤ Φi a.e.

The proof of the speedup theorem is traditionally split in two
parts, proving first a slightly simpler pseudo speedup theorem
where we do not require that the function f ′ = ϕj computed
by the faster program is precisely f but that it is just almost equal
to it (notation: f ≈ f ′). Since the computation on a finite number
of inputs does not affect the asymptotic complexity of functions,
we obtain the speedup theorem as a simple corollary.

In this article, we shall merely focus on the pseudo speedup
theorem. It is probably worth to anticipate that the proof is far from
trivial: for instance, in his well known introduction to recursion
function theory, Cutland ([15], pag.219) observes that the proof of
this theorem is probably the most difficult in this book.

Another important point to remark is that we shall not work
with Blum’s abstract framework, simply because it is not suitable
for a formal development of complexity theory. The point is that
Blum’s “axioms” do not provide a real axiomatization, since they
rely on the delicate notion of computable function. In particular, the
fact that ϕ is a principal effective enumeration (see e.g.[22]) of all
partial recursive functions is used in an essential way in most proofs
based on Blum’s axioms, usually by an invocation of Church’s
Thesis. This notion is quite difficult to express in formal terms,
and would require an early commitment to a specific computational
model. Our methodology, instead, was to start from an analysis of
the algorithms involved in the proof to derive, through a reverse
engineering process, the natural model of computation inside which
we could comfortably work at a formal level.

Essentially, we shall preserve as much as possible of Blum’s ab-
stract framework (that is just an alternative way to express Kleene’s
T predicate), dropping the requirement thatϕ is a principal enumer-
ation of all recursive functions, and adding suitable closure condi-
tions as required by the algorithms in the proof. As we shall see, in



the case of the speedup theorem, the most natural framework seems
to be provided by the class of primitive recursive functions.

As a byproduct of the proof we obtained an axiomatic frame-
work for expressing and proving complexity properties of primitive
recursive functions that seems to have an interest in its own.

The formalization has been conducted with the assistance of
the Matita Interactive Theorem Prover [7, 8]. Matita is based,
similarly to Coq, on the Calculus of Inductive Constructions; in
particular, its constructive nature helped us to emphasize a few
delicate issues in the proof of speedup theorem, discussed in the
article (see Section 4.3 and the conclusions).

In the following sections, all parts inside frames are excerpts
of Matita scripts. The whole development is available at http:
//www.cs.unibo.it/~asperti/speedup.tar.

In the current state of the development we largely (ab)used
axioms as a way to declare the abstractions parametrizing the proof.
In the end, when the axiomatic framework will become more stable,
we plan to pack axioms inside suitable “algebraic” theories.

2. Outline of the proof
In this section, we give a quick outline of our proof of the speedup
theorem. It is worth to stress again that our approach is an original
revisitation of the traditional proofs, that is not only more suitable
for a formal development, but also sheds more light on the actual
complexity of the involved algorithms. We shall make a comparison
with the customary approach in the conclusion of this article.

Let ϕi be an enumeration of (computable) functions. We shall
write {i � x} ↓ t to express the fact that program i terminates its
computation on input xwith resource bound t (that is an intensional
property of program i, and not an extensional property of ϕi).
Let h be a binary computable function; we define a family ghi (x)
of functions such that

• (cond.1) ghi (x) = gh0 (x) almost everywhere

• (cond.2) if gh0 (x) = ϕi, then for no x > i, {i�x} ↓ h(i+1, x),

Moreover, we shall also prove that

• (cond.3) for any r, there exists hr such that the complexity of
computing gr◦h

r

i (x) is less than hr(i, x).

Then, we are done. Suppose indeed that f = gr◦h
r

0 is com-
puted by some program ϕi. Then, by (cond.2) the complexity of
ϕi(x) is definitely larger than r(hr(i+ 1, x)), but by (cond 3.) and
(cond.1) gr◦h

r

i+1 (x) computes an almost equal function with com-
plexity hr(i+1, x) (as already pointed out by Young [27], here we
are making an implicit use of the smn theorem).

The rest of the paper is devoted to the definition of ghi (x) and
the proof of conditions 1-3.

3. Basic Framework and notation
The starting point of our axiomatization is Kleene’s T predicate,
that is the decidability of bounded interpretation. Since we work in
a constructive setting, this can be simply achieved by axiomatising
the existence of a function U with the following type:� �
axiom U: nat → nat → nat→ option nat .� �
The intuitive meaning is that

U i x r =


Some y if program i on input x returns y

with resource bound r
None otherwise

The only assumption we make about U is that it is “monotonic”
with respect to the amount of resources at our disposal:

� �
axiom monotonic U: ∀i,n,m,y. n ≤m→

U i x n = Some ? y →U i x m = Some ? y.� �
We recall that the question mark, in Matita, stands for an implicit
parameter, that is a term that the type inference algorithm should
be able to infer by itself. Similarly . . . can be used to express an
arbitrary number of implicit parameters.

From the previous axiom we easily conclude that U is single
valued:� �
lemma unique U: ∀i,x,n,m,yn,ym.

U i x n = Some ? yn→U i x m = Some ? ym→ yn = ym.� �
We say that the computation of program x on input y terminates
with resource bound r (notation: {i�x} ↓ r) if there exists y such
that U i x r = Some y:� �
definition terminate :=λi ,x, r . ∃y. U i x r = Some ? y.� �
It is straightforward to prove that the previous notion of (bounded)
termination is decidable:� �
lemma terminate dec: ∀i ,x,n. {i � x} ↓ n ∨¬ {i � x} ↓ n.� �
In order to define the family of functions g, we need a boolean
version of the termination test:� �
definition termb :=λi ,x, t .

match U i x t with [None ⇒ false |Some y ⇒ true].� �
It is easy to prove that termb reflects terminate in the sense of
[16]:� �
lemma termb true to term : ∀i ,x, t .

termb i x t = true →{i � x} ↓ t .

lemma term to termb true : ∀i ,x, t .
{i � x} ↓ t → termb i x t = true .� �

It is also convenient to have a function returning the result of
computation as a natural number instead of an option� �
definition out :=λi ,x, t .

match U i x t with [ None ⇒ 0 | Some z ⇒ z].� �
Given a “partial” function f : nat → option nat we say that i is a
code for f if U i x is definitely equal to f x:� �
definition code for :=λf, i .∀x.∃n.∀m. n ≤m→U i x m = f x.� �
Let us also observe that we can always regard a total function
f : nat→ nat as a partial function of type nat→ option nat via the
following, obvious transformation:� �
definition total :=λf.λx:nat . Some nat ( f x ).� �
To conclude this section, it is worth to observe that, as weak as it
can appear, this basic framework is already sufficient to prove the
gap theorem of computational complexity [2, 12].

4. The family g

The idea behind the definition of the function ghu(x) is to make
it different from any function ϕi, u ≤ i < x, such that the
computation ϕi(x) terminates with complexity h(i + 1, x) (but it
doesn’t for any input smaller than x, i.e. i has not been “cancelled”
already). This is enough to ensure condition 2 of the outline, and
condition 1 will follow easily. For condition 3, we shall need to
study the complexity of a program computing g.

http://www.cs.unibo.it/~asperti/speedup.tar
http://www.cs.unibo.it/~asperti/speedup.tar


4.1 Big operators
The definition of the speedup function involves bounded minimiza-
tion µ and the computation Max of a maximum element in a given
range. Both are simple examples of so called big-operators [9].

A big operator is a higher-order construct iterating a function F
over all elements in a given range, and combining the results with
an operator op. A default value nil is returned when the range is
empty.

Matita’s notation for big operators has the following shape (see
[5] for more details):� �
\big[op, nil ] { range description } F� �

In this article, we shall mostly work with numerical ranges of the
form i ∈ [a, b] or i ∈ [a, b[ where a and b are the lower and upper
bound of an interval of natural numbers (with the upper bound b
respectively included and excluded from the range; similarly for
the lower bound). The variable i may occur free in F , and is bound
by the operator. The range can be further restricted specifying an
additional boolean predicate, acting as a filter.

For instance, minimization can be expressed as a big operator
where we combine elements in a given range enjoying a filter
predicate p via the binary minimum operator min. In particular,
we adopt the following notations:� �
µ {i ∈ [a ,b] | p i} = \big[min,S b] {i ∈ [a ,b] | p i} i .

Max {i ∈ [a ,b] | p i}(f i ) = \big[max,0] {i ∈ [a ,b] | p i}(f i ).� �
4.2 The formal definition of g
Using minimization and Max we can give a very simple, formal
definition of the speedup function:� �
definition min input :=λh,i ,x.
µ {y ∈ ] i ,x] } (termb i y (h (S i ) y )).

definition g :=λh,u,x.
S (Max {i ∈[u,x[ | min input h i x = x} (out i x (h (S i ) x ))).� �

In the literature, if min input h i x = x it is customary to say that
i is cancelled at stage x; the range of Max in g is hence the set

Ch
u(x) = {i ∈ [u, x[ |i is cancelled at stage x}

While the explicit definition of Ch
u(x) definitely helps in the dis-

cursive exposition of the proof, at the formal level it is better to
implicitly describe it via its boolean characteristic function, used in
the range of big operators.

4.3 Basic properties of ghu
For any x ≤ u the set Ch

u(x) is empty, so ghu(x) = 1:� �
lemma le u to g 1 : ∀h,u,x. x ≤ u→ g h u x = 1.� �
The first relevant fact is the following cancellation property:

for any u, there exists an index nu such that, for any x > nu

no i < u belongs to Ch
0 (x).

Indeed, consider an i in the interval [0, u[. If, for some x, ϕi(x) ter-
minates with complexity h (S i)x then there will exists a minimum
input ni verifying such a property; if we define nu as the maximum
of all ni, i will be eventually cancelled before nu and hence cannot
belong to Ch

0 (x).
The important corollary is that for x > nu, Ch

0 (x) = Ch
u(x) and

hence
gh0 (x) ≈ ghu(x)

that is our first condition on g.
Unfortunately, the previous proof of the cancellation property

is not constructive: the search for a minimum ni such that ϕi(ni)
terminates with complexity h (S i)ni can potentially diverge and
no upper bound can be provided in advance.

Luckily, we can content ourselves with a slightly weaker prop-
erty (that is essentially a double negation variation of the previous
one): what we can prove constructively is that, for any u the fol-
lowing property is absurd:

P (u) = ∀nu∃x > nu∃i < u, i ∈ Ch
0 (x)

The proof is by induction on u. If u = 0 the result is trivial
since we have no i < 0. At the inductive step, we need to prove
that ¬P (u) → ¬P (u + 1), that by contraposition reduces to
P (u + 1) → P (u). Take nu; by P (u + i) instantiated with
nu+1 = nu, we know that there exists x > nu such that for some
i < S u, i ∈ Ch

0 (x). If i < u we are done; if i = u we apply again
P (u+ 1) starting from nu+1 = x, that is enough to cancel i too.

Formally, the property can be expressed in the following form:� �
lemma eventually cancelled : ∀h,u.¬∀nu.∃x.
nu < x ∧
max {i ∈ [0, u[ | min input h i x = x} (out i x (h (S i ) x)) 6= 0.� �

We must also revisit the notion of being almost equal. In a con-
structive setting, the basic relation is not equality but apartedness,
and two objects are defined to be equal when they cannot be taken
apart. Accordingly, two objects are almost equal when it is false
that they can be eventually taken apart:� �
definition almost equal :=λf,g:nat → nat.
¬∀nu.∃x. nu < x ∧ f x 6= g x.� �

Henceforth, we shall use the notation f ≈ g to express the fact
that two functions are almost equal in the sense of the previous
definition.

Using lemma eventually_cancelled and basic decomposi-
tion properties of big operators it is then easy to prove our first
condition for g:� �
lemma condition 1: ∀h,u.g h 0 ≈ g h u.� �
Let us now tackle the second condition:� �
lemma condition 2: ∀h,i .

code for ( total (g h 0)) i →¬∃x. i<x ∧{i � x} ↓ h (S i ) x.� �
Suppose i is a program computing gh0 , and let us suppose that there
exists x > i such that ϕi(x) terminates with complexity h (Si)x.
Then, there must exist a minimum x0 satisfying this property,
that implies that i ∈ Ch

0 (x0). By definition of g, we must have
gh0 (x0) > ϕi(x0), contradicting the fact that i was a program for
gh0 . The formal proof is not much longer, and essentially analogous.

5. The complexity of g, informally
Let us consider again the definition of g:� �
definition min input :=λh,i ,x.
µ {y ∈ ] i ,x] } (termb i y (h (S i ) y )).

definition g :=λh,u,x.
S (max {i ∈[u,x[ | min input h i x = x} (out i x (h (S i ) x ))).� �

The complexity of computing g is bound by the sum, for u ≤ i <
x, of the complexity of computing (min input h i x) plus the com-
plexity of computing (out i x (h (S i) x)). In turn, the complexity of



(min input h i x) is bound by the sum, for i < y ≤ x, of the com-
plexity of (termb i y (h (S i) y)).

If h is a constructible function (see section 6.3), the complexity
of computing both (out i x (h (S i) x)) and (termb i x (h (S i) x)) is
bound by the complexity of running an interpreter for i on input x
with a resource bound of (h (S i) x): let us call sU(i,x,h (S i) x) such
a complexity.
So, the complexity of ghu(x) does only depend on the sum of the
values of sU(i,y,h (S i) y) for u ≤ i < x and i < y ≤ x (see
Figure 1).

g  (x)
u

  i < y < x

x
     sU(i,y,h (S i) y)

u  

  u < i < x

Figure 1. Complexity of ghu(x) w.r.t. sU(i, y, h (S i) y)

The complexity of ghu(x) has hence a behaviour similar to that
described in (see Figure 2): the complexity of ghu(x) is increasing
in x and decreasing in u. At the end, we are interested to instantiate

x

u

Figure 2. Complexity behaviour of ghu(x)

h with an upper bound to the complexity ghu(x) (since g itself is
not recursive, this will not be difficult).
We may assume that sU is monotonic in all its arguments1. Sup-
posing h is antimonotonic in its first argument and monotonic in
the second one, we obtain the following upper bound for the com-
plexity of ghu(x):

∑
u≤i<x

∑
i<y≤x

sU(i, y, h (S i) y) ≤ (x− u)2 · sU(x, x, h (S u)x)

(1)

1 Recall that sU is a complexity function: since we are interested in the
asymptotic behaviour, we may e.g. suppose that, on in input x, it is defined
as the maximum complexity for all input less or equal to x.

Let now r be an arbitrary, monotonic, increasing and constructible
function, and let us consider the following recursive function:

hr(u, x) =

{
1 if x ≤ u
(x− u)2 · sU(x, x, r(hr(u+ 1, x))) otherwise

Then, by equation 1 the complexity of gr◦h
r

u (x) is less than

(x− u)2 · sU(x, x, r(hr (S u)x)) = hr(u, x)

that is our third condition.
The previous proof has been slightly simplified for the sake of

clarity. The point that was somewhat overlooked is the complexity
of ghu(x) when u ≤ x, that we assumed to be constant. While
this is sensible for certain complexity measures such as space, for
others it can depends on inputs. This is for instance the case for
time complexity, where we usually require the program to consume
its inputs, and hence to have at least a complexity proportional to
their dimension.

In our formal framework we shall assume to have a minimal
structural complexity MSC expressing this cost. Intuitively, you can
think of MSC as the complexity of the identity program; typically,
all basic primitive operations will be assumed to have this complex-
ity.

In the case of time, MSC is essentially a size measure, that is a
logarithmic function; in the case of space, MSC ∈ O(1). Instead
of fixing a specific measure, however, it is more interesting to look
for the abstract properties required for this function.

6. The formal complexity framework
6.1 Pairing
It is convenient to have at our disposal a primitive method for
packing together multiple outputs into a single one. This means
we need a pairing function mapping two natural numbers a and b
into a natural number 〈a, b〉. We assume pairing is a bijection, with
projections called fst and snd .

For instance, using pairs we can naturally rephrase U in the
following way, avoiding the option in output:� �
definition pU : nat → nat→ nat→ nat :=
λi ,x, r .〈termb i x r , out i x r〉 .

lemma pU vs U Some : ∀i,x,r,y.
pU i x r = 〈1,y〉 ↔ U i x r = Some ? y.

lemma pU vs U None : ∀i,x,r.
pU i x r = 〈0,0〉 ↔ U i x r = None ?.� �

We shall also use the pairing function to avoid working with n-
ary functions, that would be an annoying complication in a formal
setting.

The existence of the pairing function and its properties are
assumed axiomatically. The reader may wonder why we did not
give a more concrete definition. The point is that later on (see
Section 6.5) we shall be forced to make complexity assumptions
on pairing operations, and it does not make much sense to attribute
an abstract complexity measure to a concrete encoding.

6.2 Complexity Classes
We shall define complexity classes in terms of the asymptotic be-
havior of functions, so we need a small library of results dealing
with the traditional operators (big-O, small-o, etc.) used to charac-
terize functions according to their growth rates. In this article, we
shall only use the big-O operator, defined as follows:� �
definition O: relation (nat→ nat) :=λf,g.
∃c.∃n0.∀n. n0 ≤ n→ g n ≤ c∗ ( f n ).� �



It is easy to develop a small library of results, expressing the typical
properties of these operators.

A delicate issue in the formalization of Complexity Theory is
the choice between expressing complexity in terms of inputs or in
terms of their size. We already investigated the latter approach in
[3]; also in order to compare the two approaches we shall follow
the first route in this article 2.

For any complexity function s, we define the complexity class
C s as the collection of all programs that terminate their computa-
tion in O(s):� �
C :=λs, i .∃c.∃a.∀x.a ≤ x→∃y. U i x (c∗(s x)) = Some ? y.� �

Similarly, we say that a total function f : nat → nat is in CF s
if there exists a program i ∈ C s such that i is a code for (total f)
(see Section 3 for the definition of total).� �
definition CF :=λs,f .∃i . code for ( total f ) i ∧C s i .� �
If a function f is computable inO(h) we expect that both the input
x and the output f(x) have a minimal structural complexity bound
by O(h) as well:� �
axiom MSC in: ∀f,h. CF h f →∀x. MSC x ≤ h x.
axiom MSC out: ∀f,h. CF h f →∀x. MSC (f x) ≤ h x.� �
Moreover, we assume MSC is monotonic, less than the identity,
and distributes over pairing (this latter axiom can be probably
spared with)� �
axiom MSC : nat→ nat.
axiom MSC le: ∀n. MSC n ≤ n.
axiom monotonic MSC: monotonic ? le MSC.
axiom MSC pair: ∀a,b. MSC 〈a,b〉 ≤MSC a + MSC b.� �
It is easy to prove that CFs is extensional and monotonic:� �
lemma ext CF : ∀f ,g,s . (∀n. f n = g n) →CF s f →CF s g.
lemma ext CF l: ∀f ,s1 ,s2 . (∀n. s1 n = s2 n) →CF s1 f→CF s2 f.
lemma O to CF: ∀s1,s2,f . O s2 s1 →CF s1 f→CF s2 f.� �
6.3 Constructibility
A function is said to be constructible w.r.t some complexity mea-
sure when the complexity of the computation is manifest in the
(size of the) result. More formally, a function f is called space-
constructible if there is a program that given an input of length n
returns an output of length f(n) running inO(f(n)) space; the def-
inition of time-constructible function is analogous, changing space
with time.

Avoiding to talk about the size of inputs, the formal definition is
even simpler: f is constructible if, for all n, f(n) can be computed
in O(f(n)).� �
definition constructible :=λs. CF s s .� �

Constructible functions play an essential role when used as
bounds for running an interpreter: the interpretation can be run in
time O(f) only if the bound itself can be computed in O(f).

A typical example of a function that is not time constructible
is the logarithmic function, since the time required for reading
the input is already linear in its dimension. Similarly no decision
algorithm is computed by a constructible function.

2 It is worth to observe that the choice between the two paradigms is also
related to the issue of the arity of functions. The point is that from the
size of 〈a, b〉 we can deduce nothing interesting about the size of a and
b. Expressing complexity in terms of the size of inputs has, among others,
the disadvantage to force working with n-ary functions.

It is worth to observe that given any computable function f we
may always find a computable function g such that f ∈ O(g) and g
is constructible. In particular, the complexity order of a computable
function is usually expressed by a constructible bound.

These are a couple of simple lemmas we may prove on con-
structible functions.� �
lemma constr comp : ∀s1,s2 . constructible s1 → constructible s2 →

(∀x. x ≤ s2 x) → constructible (s2 ◦ s1 ).

lemma ext constr : ∀s1,s2 . (∀x.s1 x = s2 x) → constructible s1 →
constructible s2 .� �
6.4 Smn-theorem
The Smn-theorem of computability theory (see [22]) says that any
instance of a computable function obtained by fixing some input n
is still computable, and that a program computing the instance can
be found effectively as a function n.

It is natural to extend the theorem taking complexity into ac-
count. As a matter of fact, the complexity of the instance cannot be
larger than the complexity of the source program (see e.g. [1, 27]).

In our framework, this can be stated in a particularly simple and
elegant way:� �
axiom smn: ∀f,s . CF s f →∀x. CF (λy.s 〈x,y〉) (λy.f 〈x,y〉 ).� �
6.5 Complexity of primitive constructs
We shall assume the following complexities for the primitive struc-
tural operations:� �
axiom CF id: CF MSC id.
axiom CF comp fst: ∀h,f . CF h f →CF h (fst ◦ f ).
axiom CF comp snd: ∀h,f. CF h f →CF h (snd ◦ f ).
axiom CF comp pair: ∀h,f ,g. CF h f →CF h g→CF h (λx. 〈f x,g x〉 ).
axiom CF eqb: ∀h,f ,g. CF h f →CF h g→CF h (λx.eqb (f x) (g x )).� �
For composition we have� �
axiom CF comp: ∀f,g,sf , sg ,sh . CF sg g →CF sf f →

O sh (λx. sg x + sf (g x)) →CF sh (f ◦ g ).� �
6.6 Primitive Recursion
The computations of µ and Max are based on big operators, that
are a particular case of primitive recursion. The primitive recursion
scheme is defined in the following way, where m should be under-
stood as a vector of parameters):� �
let rec prim rec (k,h:nat → nat) n m on n :=

match n with
[ O ⇒ k m
| S a ⇒ h 〈a,〈prim rec k h a m, m〉〉].

definition unary pr :=λk,h,x. prim rec k h ( fst x) (snd x ).� �
Supposing that k can be computed inO(sk) and h can be computed
in O(sh) we expect to be able to compute unary pr k h) with a
complexity, in time, that grows as the following function:� �
let rec prim rec compl (k,h,sk ,sh: nat → nat) n m on n :=

match n with
[ O ⇒ sk m (∗ cost of k ∗)
| S a ⇒ prim rec compl k h sk sh a m + (∗ cost of rec . call ∗)

sh 〈a,〈prim rec k h a m, m〉〉]. (∗ cost of h ∗)� �
This can be stated by the following axiom, expressed in unary
format:



� �
axiom CF prim rec gen: ∀k,h,sk ,sh ,sh1. CF sk k →CF sh h→

O sh1 (λx. fst (snd x) +
sh 〈 fst x,〈unary pr k h 〈 fst x,snd (snd x)〉 , snd (snd x)〉〉) →

CF (unary pr sk sh1) (unary pr k h ).� �
6.7 Arithmetic primitives
Primitive recursion provides a convenient way to express the flow
structure of iterative programs, but we cannot write efficient arith-
metic programs without resorting to an efficient representation of
integers. Even in this case, the efficiency of primitive recursive al-
gorithms suffer by well know limitations [14, 19], mostly due to the
exceeding sequentialization imposed by high-level programming
constructs.

For this reason, we shall hence suppose to have at our disposal
a sufficiently large set of arithmetic primitives, computable with
minimal structural complexity:� �
axiom CF compS: ∀h,f. CF h f →CF h (S ◦ f ).
axiom CF le: ∀h,f ,g. CF h f →CF h g→CF h (λx. leb ( f x) (g x )).
axiom CF eqb: ∀h,f ,g. CF h f →CF h g→CF h (λx.eqb (f x) (g x )).
axiom CF max: ∀h,f,g. CF h f →CF h g→CF h (λx. max (f x) (g x )).
axiom CF plus: ∀h,f ,g. CF h f →CF h g→CF h (λx. f x + g x ).
axiom CF minus: ∀h,f,g. CF h f →CF h g→CF h (λx. f x − g x).� �
In order to prove the constructibility of the complexity bound of
the function h, we shall also need the arithmetical product. This
function cannot be computed with minimal structural complexity;
we assume the usual square bound:� �
axiom CF times: ∀f ,g,h. CF h f →CF h g→

CF (λx.h x + MSC (f x) ∗ MSC (g x)) (λx.f x ∗ g x ).� �
By the previous axiom, it is easy to obtain the following result, that
is usually simpler to use:� �
lemma CF times1: ∀f,g, sf , sg . CF sf f →CF sg g→

CF (λx.sf x ∗ sg x) (λx.f x ∗ g x ).� �
6.8 If then else
Primitive recursion embeds definition by cases, and hence the “if
then else” construct.

Exploiting the obvious coercion form bool to nat, we can easily
prove the following lemma:� �
lemma if prim rec : ∀b:nat → bool. ∀f ,g:nat → nat.∀x:nat .

if b x then f x else g x = prim rec g ( f ◦ snd ◦ snd) (b x) x.� �
More interestingly, our complexity assumptions for primitive re-
cursion allow us to derive the following complexity bound for the
if then else:� �
lemma CF if: ∀b:nat → bool. ∀f ,g,s . CF s b →CF s f →CF s g→

CF s (λx. if b x then f x else g x ).� �
The proof is not complex, but requires a clever exploitation of the
fact that the result of b is bounded.

6.9 Minimization and Max
Using primitive recursion we can also encode all big operators,
and in particular the operations of minimization and Max of Sec-
tion 4.1. As in the case of the “if then else”, we can then derive com-
plexity bounds for these constructs in our axiomatic framework. In
particular, we have been able to prove the following results:

� �
lemma CF max: ∀a,b.∀p:nat→ bool.∀f,ha,hb,hp,hf,s .
CF ha a →CF hb b→CF hp p→CF hf f→

O s (λx.ha x + hb x +
(
∑
{i ∈[a x ,b x[ }

(hp 〈 i ,x〉 + hf 〈 i ,x〉 + max {i ∈ [a x, b x [ }(hf 〈 i ,x〉 )))) →
CF s (λx.max {i ∈[a x,b x[ | p 〈 i ,x〉 }(f 〈 i ,x〉 )).� �� �

lemma CF mu: ∀a,b.∀f:nat→ bool.∀sa,sb,sf, s .
CF sa a →CF sb b→CF sf f →
O s (λx.sa x + sb x +∑

{i ∈[a x ,S(b x)[ }
( sf 〈 i ,x〉 + MSC 〈b x − i,〈S(b x ), x〉〉 )) →

CF s (λx.µ {i ∈[a x,b x] }(f 〈 i ,x〉 )).� �
In both cases, it is convenient to consider more relaxed but simpler
bounds. In our proof of the speedup theorem, we exploited the
following variants:� �
lemma CF max2: ∀a,b.∀p:nat→ bool.∀f,ha,hb,hp,hf,s .

CF ha a →CF hb b→CF hp p→CF hf f→
O s (λx.ha x + hb x +

(b x − a x)∗max {i ∈ [a x, b x [ }(hp 〈 i ,x〉 + hf 〈 i ,x〉 )) →
CF s (λx.max {i ∈[a x,b x[ | p 〈 i ,x〉 }(f 〈 i ,x〉 )).� �� �

lemma CF mu4: ∀a,b.∀f:nat→ bool.∀sa,sb,sf,s . (∀x.sf x > 0) →
CF sa a →CF sb b→CF sf f →
O s (λx.sa x + sb x +

(S(b x) − a x)∗Max {i ∈[a x ,S(b x)[ }(sf 〈 i ,x〉 )) →
CF s (λx.µ {i ∈[a x,b x] }(f 〈 i ,x〉 )).� �

6.10 Complexity of bounded interpretation
The final ingredient is the possibility to perform bound interpreta-
tion inside the system, that essentially amount to the internalization
of the function U .

More precisely, let us consider the unary version of the function
pU of Section6.1:� �
definition pU unary :=λp. pU ( fst p) ( fst (snd p)) (snd (snd p )).� �
We simply require that pU is computable with some complexity
sU .� �
axiom sU : nat → nat.
axiom CF U : CF sU pU unary.� �
We suppose sU to be mononotic in all its arguments (as we al-
ready observed, this is a natural requirement for any complexity
function).� �
axiom monotonic sU: ∀i1,i2 ,x1,x2,s1 ,s2 .

i1 ≤ i2 → x1 ≤ x2→ s1 ≤ s2→
sU 〈i1 ,〈x1,s1〉〉 ≤ sU 〈i2 ,〈x2,s2〉〉 .� �

Moreover, we expect sU to grow more than linearly in the resource
bound:� �
axiom sU le: ∀i ,x,s . s ≤ sU 〈i ,〈x,s〉〉 .� �
It is easy to prove that sU is also the complexity of the (unary
versions of) termb and out :� �
lemma CF termb: CF sU termb unary.
lemma CF out: CF sU out unary.� �



7. The complexity of g
We have now all the machinery we need to tackle the formal
analysis of the complexity of the speedup function. The details
are not very interesting: what was interesting was the backward
reconstruction process that, with the assistance of an interactive
prover, allowed us to build the previous axiomatic framework from
the problem we were meant to solve. The current structure of the
development still reflects this small step backward activity3.

The main result is the following. Let us consider the unary
version of g:� �
definition unary g :=λh.λux. g h ( fst ux) (snd ux).� �
Then, supposing that h is a constructible binary function antimono-
tonic in its first argument and monotonic in the second one, the
complexity of unary g is sg, where sg satisfies the following equa-
tion (c is just a function sufficiently large to cover the complexity
of structural operations):� �
definition c :=λx.(S (snd x−fst x))∗MSC 〈x,x〉.

definition sg :=λh,x.
let a := fst x in
let b :=snd x in
c x + (b−a)∗(S(b−a))∗sU 〈x,〈snd x,h (S a) b〉〉 .

lemma compl g11 : ∀h.
constructible (λx. h ( fst x) (snd x)) →

(∀n. monotonic ? le (h n)) →
(∀n,a ,b. a ≤ b→ b ≤ n→ h b n ≤ h a n) →

CF (sg h) (unary g h ).� �
8. Closing the argument
To close the argument, we need to instantiate the function h of g
with a constructible upper bound of its complexity; more precisely,
we are looking for (a constructible upper bound to) a fix point of
sg.

We can easily define such a function using primitive recursion;
the bound we provide is not particularly tight; we just aimed to a
function for which we could easily prove constructibility. Here is
the actual definition:� �

let rec h of aux ( r : nat → nat) (d,b:nat ) on d : nat :=
match d with
[ O ⇒ 〈〈b,b〉,〈b,b〉〉
| S d1 ⇒ let c :=〈d1,〈h of aux r d1 b,b〉〉 in

(S c)∗(S(S c))∗sU 〈c,〈b, r (h of aux r d1 b)〉〉 ].

definition h of :=λr,p.
let m :=max (fst p) (snd p) in
h of aux r (snd p − fst p) m.� �

It is easy to prove that, if r is a monotonic function and for any n,
n ≤ r(n), then (h of r) is antimonotonic in the first argument and
monotonic in the second one. Moreover, if r is constructible, (h of
r) is constructible too.

Putting together all these results, it is easy to conclude the
following lemma:� �
lemma speed compl: ∀r:nat → nat.

(∀x. x ≤ r x) →monotonic ? le r → constructible r →
CF (h of r ) (unary g (λi ,x. r (h of r 〈 i ,x〉 ))).� �

Using smn, we get as a corollary that

3 So, it can appear a bit strange at first glance.

� �
lemma speed compl i: ∀r: nat → nat.

(∀x. x ≤ r x) →monotonic ? le r → constructible r →
∀i . CF (λx.h of r 〈 i ,x〉) (λx.g (λi ,x. r (h of r 〈 i ,x〉 )) i x ).� �

We can now prove the pseudo speedup theorem that we state in the
following way: for any function r that is monotonic, constructible
and for any n, n ≤ r(n), there exists a function f such that if the
complexity of f is sf , there exists another function f ′ such that
f ′ ≈ f and f ′ has a complexity sf ′ where r ◦ sf ′ ∈ O(sf ):� �
theorem pseudo speedup:
∀r: nat → nat. (∀x. x ≤ r x) →monotonic ? le r → constructible r →
∃f .∀sf . CF sf f →∃f’,sf ’. f ≈ f’ ∧CF sf’ f ’ ∧O sf ( r ◦ sf ’).� �

For the proof, we just take

f = g (λi, x.r (h of r 〈i, x〉)) 0)

and
f ′ = g (λi, x.r (h of r 〈i, x〉)) (S i)

By lemma speed compl i the complexity of f ′ is then

sf ′ = λx.h of r 〈S i, x〉

and the proof that r ◦ sf ′ ∈ O(sf) is trivial.

9. Conclusions
Our proof has been inspired by the simplified version of Blum’s
proof proposed by Young [27]. The main difference is that in
Young’s proof the function gu(x) is defined in terms of its own
complexity; quoting from [27]:

We will also assume that it is legitimate to define a function
recursively, not just from its earlier values, but also from its
earlier run times. Intuitively, this amounts to assuming that
if we used a program to calculate the value of a function
at an early argument, we can know the resources used
in the computation even if we do not explicitly know the
entire program used for computing the function. Formally of
course, one must use the recursion theorem or some other
means to validate such an argument.

A more formal version of Young’s argument, making an explicit use
of the recursion theorem, can be found in [15] (see also [25]). How-
ever, the recursion theorem is a quite heavy tool of computability
theory, imposing - among other things - to work in a general recur-
sive setting.
The proof presented in this article is based on two key observations:

• we do not really need to work with the actual complexity of
gu(x): any upper bound h to such a complexity will do the job;
• we can abstract the definition of gu(x) w.r.t. this function h,

and instantiate it later according to the complexity of g.

This approach has several advantages. First of all, it makes no
use of the fixed point theorem, hence providing information about
speedup phenomena in subrecursive settings. For instance, if r is
primitive recursive, gr◦(h

r)
0 is too, hence primitive recursive func-

tions do not have, in general, a inherent computational complexity.
Up to our knowledge, the only other article addressing a similar
problem is [18], where the result is only stated (with no proof) in
terms of space complexity, referring to the Ph.D. Thesis of Ritchie
[21] for the complexity analysis of the Turing machine implement-
ing the speedup function.

Another technical advantage of our proof is that the termination
of gu(x) is not an issue, while it becomes delicate when making use
of fix points (in [15], termination takes a good part of the proof).



The price to pay is a detailed investigation of the complexity
of gyu(x); however, this provides interesting information on the
complexity of the functions that can be effectively sped-up (in
particular, one could still wonder about speedup phenomena below
the complexity of gyu(x)). Finally, our development (in contrast
with [18]) proves that this complexity analysis can be done at a
comfortable level of abstraction, avoiding the need to dig with
Turing Machines.

The work presented in this article is part of a large program of
formal revisitation of Complexity Theory aiming to a synthetic sys-
tematization of the field, particularly oriented to machine verifica-
tion. Our approach is based on a reverse methodology [3], aiming at
reconstructing from proofs of known results in Complexity Theory
the basic, abstract notions and assumptions underlying them; the
approach can be essentially compared with the work of defining an
application programming interface starting from the investigation
of the set of services that are supposed to be offered by the appli-
cation.

The methodology has been already applied to the hierarchy
theorems [3], and the gap theorem [2]. The speedup function,
discussed in this article, was a particuarly challenging test-bench,
due to the complexity of the algorithms involved in the proof.

A minor issue still deserves a more detailed investigation. Us-
ing our complexity results for the if-then-else, and in particular the
CF if lemma of section 6.8, it is easy to prove that any pair of al-
most equal functions have the same complexity (and hence that the
pseudo-speedup theorem entails the speedup theorem). However,
if we start from the weak notion of “almost equal” of section 4.3,
there seems to be no way to provide a constructive proof of the pre-
vious result. On the other side, as we already discussed, working
with the traditional notion, we did not see a way to constructively
prove the pseudo-speedup theorem. In conclusion, the whole proof
of the speed-up theorem, as far as we can see at present, does not
appear to be constructive.

Possible extensions of the work presented in this article con-
sist in studying the possibility to decompose the U function in a
sequence of more elementary transition steps (along the lines of
[13]); this seems an important preliminary step to define the reach-
ability graph among configurations, that plays a major role in many
important results of Complexity Theory such as the theorems of
Savitch [23] and Immerman-Szelepcsényi [17, 24].

Finally, our axiomatic framework must be eventually validated
by some concrete computational model. Our syntetic approach
naturally complements the more traditional, concrete approach to
(formal) computability and complexity, starting from the definition
and analysis of specific models of computation, that was recently
initiated by many different authors [4, 6, 20, 26].

